
*=============PWM_WithOut_PM_Spectrum========================

Being able to view pure AM or pure PM spectrums makes it 
possible to see that there might be a way to perform
Pulse Width Modulation on a square wave which actually
does not introduce any Phase Modulation.

PWM_Without_PM_spec
* dsauersanjose@aol.com  2.12.10_1.26PM 
* www.idea2ic.com 
* 
* 
*     Vtime       ^  OUT 
*        ____    /_\ 
*  VT  _|_   |   _|_ 
*     /_  \  |  /_  \ 
*    // \  \ |_// \  \ 
*    \   \//   \   \// 
*     \___/     \___/ B1 
*       |         | 
*      _|_       _|_ 
*      ///       /// 
* 
VP        VP        0     DC     3.141592653589793
VT        VT        0     PWL    ( 0 0 1 1 ) 
B1        TRI       0     V =    acos(cos(2*v(VP)*20*v(VT)))/v(VP)
B2        OUT       0     V =    2*u(1-v(TRI)*2  + .5*cos(2*v(VP)*1*v(VT)) )-1
*=========Run_Simulation=========================
.tran     .1m       1  0     .1m    
.control 
run 
set       pensize = 2 
plot      v(out) 
*================================================================
In this case, a 20Hz square wave is PWM with a 1Hz sine wave. 

*===========Find_Spectrum=========================
linearize 
set         specwindow=   "none" 
*SPEC       FSTART  FSTOP    FSTEP   VECTOR
spec        1       5k        1     v(out)
set         pensize = 2
plot        dB(abs(real(v(out)))) dB(abs(imag(v(out))))   xlog ylimit -100 0

*================================================================



This simulation is designed to output most everything on the
spectrum as real.

*===========Print_Spectrum=========================
foreach ii  1 2 3 4 5 18 19 20 21 22 38 39 40 41 42 58 59 60 61 62  78 79 80 81 92 98 99 100 101 102
let i =     $ii
let fr=     frequency[i-1]
let vfftr = real(v(out)[i-1])
let vffti = imag(v(out)[i-1])
echo freq=  $&fr   real=  $&vfftr  imag=  $&vffti  
end
*================================================================

All of the sidebands are performing amplitude modulation.
Most of the PWM is being done by the second harmonic of 
the output square wave under going amplitude modulation.
And there is some 1Hz signal in the output as well.

Most of the other sidebands appear to be under going AM to
keep the square shape of the PWM output.

Circuit:  PWM_Without_PM_spec

freq = 1  real = 0.50062      imag = 7.54917E-06
freq = 2  real = -6.00044E-05 imag = -1.80969E-09
freq = 3  real = 0.000454372  imag = 2.05553E-08
freq = 4  real = 0.000881114  imag = 5.31475E-08
freq = 5  real = 0.0011215    imag = 8.45587E-08
freq = 18 real = -0.0924626   imag = -2.50969E-05
freq = 19 real = -3.18749E-05 imag = -9.13247E-09
freq = 20 real = 1.08389      imag = 0.000326885
freq = 21 real = 9.01494E-05  imag = 2.85469E-08
freq = 22 real = -0.0947299   imag = -3.14259E-05
freq = 38 real = 0.000943628  imag = 5.40676E-07
freq = 39 real = -0.361751    imag = -0.000212728
freq = 40 real = -5.32073E-05 imag = -3.20906E-08
freq = 41 real = -0.360042    imag = -0.000222579
freq = 42 real = -0.000429113 imag = -2.71747E-07
freq = 58 real = 0.179944     imag = 0.00015735
freq = 59 real = -0.000618368 imag = -5.50045E-07
freq = 60 real = -0.0102989   imag = -9.3162E-06
freq = 61 real = 0.000252182  imag = 2.31919E-07
freq = 62 real = 0.180187     imag = 0.000168424
freq = 78 real = -0.000738199 imag = -8.67956E-07
freq = 79 real = 0.0910715    imag = 0.000108451
freq = 80 real = 0.000338961  imag = 4.08754E-07
freq = 81 real = 0.0891658    imag = 0.000108868



So does this mean there is No Phase Modulation using
this method of PWM?

===========Full_Netlist_For_Copy_Paste=======================
PPWM_Without_PM_spec
* dsauersanjose@aol.com  2.12.10_1.26PM 
* www.idea2ic.com 
* 
* 
*     Vtime       ^  OUT 
*        ____    /_\ 
*  VT  _|_   |   _|_ 
*     /_  \  |  /_  \ 
*    // \  \ |_// \  \ 
*    \   \//   \   \// 
*     \___/     \___/ B1 
*       |         | 
*      _|_       _|_ 
*      ///       /// 
* 
VP          VP        0     DC     3.141592653589793
VT          VT        0     PWL    ( 0 0 1 1 ) 
B1          TRI       0     V =    acos(cos(2*v(VP)*20*v(VT)))/v(VP)
B2          OUT       0     V =    2*u(1-v(TRI)*2  + .5*cos(1*v(VP)*2*v(VT)) )-1
*===========Run_Simulation=========================
.tran       .1m       1  0     .1m    
.control 
run 
set         pensize = 2 
plot        v(out) 
*===========Find_Spectrum=========================
linearize 
set         specwindow=   "none" 
spec        1   5k        1     v(out)
set         pensize = 2
plot        dB(abs(real(v(out)))) dB(abs(imag(v(out))))   xlog ylimit -100 0
*===========Print_Spectrum=========================
foreach ii  1 2 3 4 5 18 19 20 21 22 38 39 40 41 42 58 59 60 61 62  78 79 80 81 92 98 99 100 101 102
let i =     $ii
let fr=     frequency[i-1]
let vfftr = real(v(out)[i-1])
let vffti = imag(v(out)[i-1])
echo freq=  $&fr   real=  $&vfftr  imag=  $&vffti  
end
.endc 
.end 

